

METODE NUMERICE DE REZOLVARE A ECUAȚIILOR ALGEBRICE ȘI TRANSCENDENTE

1. SEPARAREA SOLUȚIILOR ECUAȚIILOR ALGEBRICE ȘI TRANSCENDENTE

În cazul, când ecuația algebrică sau transcendentă are o structură simplă, soluțiile pot fi determinate exact și relativ ușor. Dacă însă structura ecuației este complicată, procedura de determinare a soluțiilor devine destul de anevoioasă. Mai mult decât atât, atunci când ecuația modelează careva situații, fenomene care depind de mai mulți parametri, valoarea cărora este cunoscută doar aproximativ, noțiunea de soluție exactă își pierde în general sensul. De aceea are sens de a determina careva metode de calcul aproximativ al soluțiilor ecuațiilor algebrice și transcendente.

Fie dată ecuația $f(x) = 0$ (1),

$f(x)$ fiind definită și continuă pe un careva interval $a \leq x \leq b$.

Orice valoare ξ , pentru care expresia $f(\xi) = 0$ este adevărată se numește **zerou** al funcției $f(x)$ sau **soluție** a ecuației $f(x) = 0$

În cele ce urmează se va presupune că ecuația (1) are soluții distincte (izolate), adică pentru fiecare soluție a ecuației există o vecinătate a sa, care nu conține alte soluții.

Astfel, rezolvarea unei ecuații algebrice se divide în două etape:

1. Separarea intervalelor pe care ecuația are o singură soluție și
2. Micșorarea pe cât mai mult posibil a fiecărui din aceste intervale (dacă se pune problema determinării tuturor soluțiilor) sau a unuia din ele (dacă trebuie de determinat doar una din soluții).

Pentru separarea soluțiilor se va folosi următoarea teoremă din analiza matematică:

Teoremă. Dacă funcția $f(x)$ continuă pe segmentul $[a, b]$ primește la extremitățile lui valori de semn diferit $f(a) \times f(b) < 0$ atunci pe acest segment există cel puțin un punct ξ , pentru care expresia $f(\xi) = 0$ este adevărată. Dacă pe acest segment există $f'(x)$, continuă, care are un semn constant, atunci soluția ecuației pe segmentul $[a, b]$ este unică. (fără demonstrație)

Dacă soluțiile ecuației $f(x)=0$ pot fi ușor calculate, atunci procesul de separare a soluțiilor se reduce la determinarea semnelor funcției în extremitățile segmentului $[a, b]$ și în punctele în care derivata funcției este 0. Segmentele la extremitățile cărora funcția va avea valori de semn opus vor conține câte o soluție a ecuației inițiale.

Exemplul 1. Să se separe soluțiile ecuației: $x^5 - 5x + 7$.

$$f(x) = x^5 - 5x + 7; \quad f'(x) = 5x^4 - 5.$$

Rezolvând ecuația $5x^4 - 5 = 0$ se obțin soluțiile $x = 1$ și $x = -1$.

Se verifică semnul derivatei pe intervalele:

$$(-\infty, -1): \quad f'(x) > 0;$$

$$(-1, 1): \quad f'(x) < 0;$$

$$(1, +\infty): \quad f'(x) > 0;$$

Deci, ecuația inițială va avea cel mult trei soluții, câte una pe fiecare din intervalele determinate mai sus. Urmează verificarea semnului funcției în extremitățile fiecăruia dintre intervalele stabilite:

$$(-\infty, -1): \lim_{x \rightarrow -\infty} f(x) = -\infty; \quad f(-1) = 11 \Rightarrow \exists x \in (-\infty, -1): f(x) = 0.$$

$$(-1, 1): \quad f(-1) = 11 \quad f(1) = 3 \Rightarrow \nexists x \in (-1, 1): f(x) = 0.$$

$$(1, \infty): \quad f(1) = 3 \quad \lim_{x \rightarrow \infty} f(x) = \infty \Rightarrow \nexists x \in (1, \infty): f(x) = 0$$

Prin urmare, unica soluție a ecuației se află în intervalul $(-\infty, -1)$.

Exemplul 2. Să se determine numărul de soluții a ecuației $e^x + x = 0$.

$$f'(x) = e^x + 1; \quad f'(x) > 0 \quad \forall x \in \mathbb{R}.$$

Întrucât $\lim_{x \rightarrow -\infty} f(x) = -\infty$; $\lim_{x \rightarrow \infty} f(x) = \infty$; se poate afirma că ecuația inițială are o singură soluție.

Exemplul 3. Să se separe rădăcinile ecuației $x^3 - 9x^2 + 24x - 19 = 0$ pe segmentul $[0, 8]$

$$f(x) = x^3 - 9x^2 + 24x - 19; \quad f'(x) = 3x^2 - 18x + 24.$$

Rezolvând ecuația $f'(x) = 0$ se obțin soluțiile $x_1 = 4$, $x_2 = 2$.

x	$f(x)$	Semn
0	-19	-
2	1	+
4	-3	-
8	109	+

Deci ecuația va avea trei soluții, câte una pe fiecare din segmentele $[0, 2]$, $[2, 4]$, $[4, 8]$.

Exemplul 4. Să se separe rădăcinile ecuației $x^3 - 9x^2 + 24x - 13 = 0$ pe segmentul $[0, 8]$.

$$f(x) = x^3 - 9x^2 + 24x - 13; \quad f'(x) = 3x^2 - 18x + 24.$$

Rezolvând ecuația $f'(x) = 0$, se obțin soluțiile $x_1 = 4$, $x_2 = 2$.

x	$f(x)$	Semn
0	-13	-
2	7	+
4	3	+
8	115	+

Deci ecuația va avea o singură soluție localizată pe segmentul $[0, 2]$.

Cunoașterea unui limbaj de programare ușurează mult studierea comportamentului unei funcții și în particular procesul separare a soluțiilor ecuației $f(x) = 0$. Următorul program Pascal permite vizualizarea în regim text a graficului unei funcții și separarea soluțiilor.

Prin modificarea subprogramului (funcția **f**) în care este descrisă funcția graficul căreia se construiește, programul se adaptează la oricare altă funcție acceptabilă, introdusă de utilizator:

```
program cn002;
uses crt;
const
    nmax=50;
    ymax=20;
    deplas=15;
```

```

type
    ecran=array[0..21,0..78] of char;
    valorireal=array[1..nmax] of real;
    valoriintreg=array[1..nmax] of integer;
var
    graf: ecran;
    y: valorireal;
    ynorm: valoriintreg;
    vmax,vmin,a,b: real;
    i,j,liniezero: integer;
function f(x:real):real;
begin
    f:=exp(cos(2*x)*ln(x))+3*sin(x);
end;
procedure calcul(a,b:real;var z:valorireal);
var
    h: real;
begin
    h:=abs(b-a)/(nmax-1);
    for i:=1 to nmax do z[i]:=f(a+(i-1)*h);
end;
procedure normare(var fz:valoriintreg; z:valorireal);
var
    max,delta: real;
begin
    max:=abs(z[1]); vmax:=z[1]; vmin:=z[1];
    for i:=2 to nmax do
        begin
            if abs(z[i])>max then max:=abs(z[i]);
            if z[i]>vmax then vmax:=z[i];
            if z[i]<vmin then vmin:=z[i];
        end;
    delta:=(vmax div 2) / max;
    for i:=1 to nmax do fz[i]:=round(z[i]*delta);
end;
procedure modeleazagrafic(g:ecran);
begin
    fillchar(g,sizeof(g),' ');
    for i:=deplas to nmax+deplas do
        begin
            g[liniezero,i]:='-';
            g[ymax div 2 - ynorm[i-deplas+1],i]:='*';
        end;
    for i:=0 to ymax do g[i,1]:='|';
    g[0,1]:='^';
    g[0,3]:='Y';
    g[liniezero,nmax+deplas+1]:='>';
    g[liniezero,3]:='0';
    g[liniezero,nmax+deplas+2]:='X';
    clrscr;
    for i:=0 to 21 do
        begin
            for j:=0 to 78 do write(g[i,j]);
            writeln;
        end;
    gotoxy(4,2);write(vmax:0:2);
    gotoxy(4,23);write(vmin:0:2);
    gotoxy(deplas-1,23);write ('X: ',a:0:2,' ':50,b:0:2);
    gotoxy(1,25);write('Alt interval (D)a / (N)u');

```

```


end;

begin
    repeat clrscr;
        write('Introdu extremitatile intervalului: ');
        readln(a,b);
        calcul(a,b,y);
        normare(ynorm,y);
        liniezero:= ymax div 2;
        modeleazagrafic(graf);
    until upcase(readkey)='N';
end.
    
```


Rezultate Pentru $[1, 10]$ graficul generat de program este dat în desenul 1.

Analizând graficul generat de program, se observă existența a două soluții a ecuației $x^{\cos(2x)} + 3\sin(x) = 0$ pe $[1, 10]$.


După repetarea programului pe $[2, 4]$ și $[4, 6]$ se obțin respectiv graficele din desenele 2 și 3, care indică existența soluțiilor separate pe fiecare segment. Îngustarea repetată a fiecărui segment ar permite localizarea mai exactă a soluțiilor, dar odată ce rădăcinile sunt separate, această problemă poate fi rezolvată prin metode numerice.



Desenul 1. Graficul funcției, generat de programul cn002.



Desenul 2. Detalizarea graficului funcției pe $[2, 4]$



Desenul 3. Detalizarea graficului funcției pe $[4, 6]$

Întrebări și exerciții

1. Ce numim soluție a unei ecuații?
2. Ce condiții trebuie să satisfacă funcția $f(x)$, pentru ca pe un segment dat să existe cel puțin o rădăcină a ecuației $f(x)=0$? Dar pentru existența exact a unei soluții?

Separați soluțiile ecuațiilor:

a) $y = \frac{x^4}{4} + x^3 - \frac{x^2}{2} - 3x + 8$

b) $y = 2x^3 - 6x^2 - 48x + 17$

c) $y = x^4 - 14x^2 - 24x - 4$

d) $y = e^{-x^2}$

e) $y = e^x(x^2 - 2x + 2)$

f) $y = x[\sin(\ln(x)) - \cos(\ln(x))]$

3. Elaborați un program, care calculează valorile unei funcții continue date pe un segment $[a,b]$ cu pasul h , și afișează toate intervalele, la extremitățile cărora funcția are valori cu semn opus.

2. FORMULA GENERALĂ DE DETERMINARE A ERORII*

Teoremă. Fie ξ , soluția exactă iar $\bar{x}$ - soluția aproximativă a ecuației $f(x) = 0$ pe un segment $[a, b]$, pe care $f'(x)$ are semn constant, iar m o valoare pentru care este adevărată inegalitatea:

$$|f'(x)| > m > 0 \text{ pe } [a, b].$$

În acest caz este corectă estimarea erorii de calcul $|\bar{x} - \xi|$ prin expresia $|\bar{x} - \xi| \leq \frac{f(\bar{x})}{m}$.

(în particular, drept m putem lua cea mai mică valoare a $f'(x)$ pe segmentul $[a, b]$)

Demonstrația se face în baza teoremei *Lagrange*. Conform acestei teoreme $\exists$ valoarea intermediară c între ξ și $\bar{x}$ astfel încât $f(\bar{x}) - f(\xi) = (\bar{x} - \xi) f'(c)$;

Deoarece $f(\xi) = 0$ iar $|f'(x)| \geq m$, se obține $|\bar{x} - \xi| \leq \frac{f(\bar{x})}{m}$.

■ În practică utilizarea formulei deduse mai sus dă rezultate destul de imprecise, de aceea se folosesc diverse metode de îngustare a segmentului $[a, b]$, și utilizarea în calitate de estimare a erorii a lungimii intervalului care conține atât soluția aproximativă, cât și cea exactă.

* Materialul acestui paragraf este opțional. La formula generală a erorii se face referință în paragraful 3.5 – Metoda Newton.

3. METODA BISECTIEI

Fie dată funcția $f(x)$, continuă pe segmentul $[a, b]$ și $f(a) \times f(b) < 0$.

Se cere să se determine o soluție a ecuației

$$f(x) = 0 \quad (1)$$

pe segmentul $[a, b]$. Proprietățile funcției asigură existența a cel puțin o soluție pe segmentul $[a, b]$.

Una dintre cele mai simple metode de determinare a unei soluții a ecuației $f(x) = 0$ este metoda *bisecției*. Metoda presupune determinarea punctului de mijloc c a segmentului $[a, b]$ apoi calculul valorii $f(c)$. Dacă $f(c) = 0$, atunci c este soluția exactă a ecuației. În caz contrar soluția este căutată în continuare pe acel dintre segmentele $[a, c]$ și $[c, b]$, pentru care semnul funcției în extremități este diferit (desenul 4).

Dacă $f(a) \times f(c) > 0$, atunci soluția e căutată în continuare pe segmentul $[a_1, b_1]$, unde $a_1 \leftarrow c^1$, $b_1 \leftarrow b$. În caz contrar extremitățile noului segment vor fi $a_1 \leftarrow a$, $b_1 \leftarrow c$. Procesul se reia pe segmentul $[a_1, b_1]$, repetându-se până când nu se obține soluția exactă sau (în majoritatea absolută a cazurilor!) devierea soluției calculate de la cea exactă nu devine suficient de mică.


În urma iterațiilor succesive se obține consecutivitatea segmentelor

$$[a_0, b_0], [a_1, b_1], [a_2, b_2], \dots, [a_i, b_i], \dots$$

Pentru fiecare din ele are loc relația $f(a_i) \times f(b_i) < 0$, $i=0, 1, 2, \dots$ (2)

Estimarea erorii. Deoarece ξ e un punct al segmentului $[a_i, b_i]$ rezultă că diferența dintre soluția exactă și cea calculată nu este mai mare decât lungimea segmentului $[a_i, b_i]$. Prin urmare, localizarea soluției pe un segment cu lungimea ε asigură o eroare ce nu depășește lungimea ε a segmentului.

$$|\xi - a_i| < \varepsilon = |b_i - a_i|$$



Desenul 4. Calculul consecutiv al segmentelor, care conțin soluția ecuației.

¹ Notăția $a_1 \leftarrow c$ are semnificația a_1 primește valoarea lui c

Algoritmizarea metodei**Algoritmul de calcul pentru un număr prestabilit n de aproximări succesive:****Pas 0** Inițializarea: $i \leftarrow 0$;**Pas 1** Determinarea mijlocului segmentului $c \leftarrow \frac{a+b}{2}$;**Pas 2** Reducerea segmentului ce conține soluția: dacă $f(c) = 0$ atunci soluția calculată este $x = c$. SFÂRȘIT. În caz contrar dacă $f(a) \times f(c) > 0$, atunci $a \leftarrow c$; $b \leftarrow b$, altfel $a \leftarrow a$; $b \leftarrow c$ **Pas 3** $i \leftarrow i+1$; Dacă $i = n$ atunci soluția calculată este $x = \frac{a+b}{2}$. SFÂRȘIT, în caz contrar se revine la **pas 1****Algoritmul de calcul pentru o precizie² ε dată:****Pas 1** Determinarea mijlocului segmentului $c \leftarrow \frac{a+b}{2}$;**Pas 2** dacă $f(c) = 0$ atunci soluția calculată este $x = c$. SFÂRȘIT.
în caz contrar dacă $f(a) \times f(c) > 0$, atunci $a \leftarrow c$; $b \leftarrow b$, altfel $a \leftarrow a$; $b \leftarrow c$ **Pas 3** Dacă $|b-a| < \varepsilon$ atunci soluția calculată este $x = \frac{a+b}{2}$. SFÂRȘIT, în caz contrar se revine la **pas 1****Exemplul 1:** Să se determine o rădăcină a ecuației $x^4 + 2x^3 - x - 1 = 0$ pe segmentul $[0, 1]$ pentru 16 divizări consecutive.**Program:** Deoarece numărul de aproximări succesive este fixat, iar extremitățile segmentului cunoscute, atribuirile se realizează nemijlocit în program.

```

program cn003;
var
  a,b,c: real;
  i,n: integer;

function f(x: real): real;
begin
  f:=sqr(sqr(x))+2*x*sqr(x)-x-1;
end;

begin
  a:=0; b:=1; n:=16;
  for i:=1 to n do
    begin
      c:=(b+a)/2;
      writeln('i=',i:3, ' x=',c:10:8, ' f(x)=',f(c):12:8);
      if f(c)=0 then break
      else if f(c)*f(a)>0 then a:=c else b:=c;
    end;
  end.

```

² în contextul dat precizia ε semnifică o eroare de calcul, care nu depășește valoarea ε

Rezultate

```

i= 1 x=0.50000000 f(x)= -1.18750000
i= 2 x=0.75000000 f(x)= -0.58984375
i= 3 x=0.87500000 f(x)= 0.05102539
i= 4 x=0.81250000 f(x)= -0.30393982
i= 5 x=0.84375000 f(x)= -0.13557339
i= 6 x=0.85937500 f(x)= -0.04461473
i= 7 x=0.86718750 f(x)= 0.00261236
i= 8 x=0.86328125 f(x)= -0.02114845
i= 9 x=0.86523438 f(x)= -0.00930499
i= 10 x=0.86621094 f(x)= -0.00335557
i= 11 x=0.86669922 f(x)= -0.00037392
i= 12 x=0.86694336 f(x)= 0.00111864
i= 13 x=0.86682129 f(x)= 0.00037222
i= 14 x=0.86676025 f(x)= -0.00000089
i= 15 x=0.86679077 f(x)= 0.00018565
i= 16 x=0.86677551 f(x)= 0.00009238

```

Exemplul 2: Să se determine o rădăcină a ecuației $6\cos(x) + 8\sin(x) = 0$ pe segmentul $[2, 4]$ cu precizia $\epsilon=0.00017$

```

program cn004;
var
  a,b,c,eps: real;

function f(x:real):real;
begin
  f:=6*cos(x)+8*sin(x);
end;

begin
  a:=2; b:=4; eps:=0.00017;
  repeat
    c:=(b+a)/2;
    writeln('x=',c:10:8, ' f(x)=',f(c):12:8);
    if f(c)=0 then break
    else if f(c)*f(a)>0 then a:=c else b:=c;
  until abs(b-a)<eps;
end.

```

Rezultate:

```

x=3.00000000 f(x)= -4.81099492
x=2.50000000 f(x)= -0.01908454
x=2.25000000 f(x)= 2.45554384
x=2.37500000 f(x)= 1.22780943
x=2.43750000 f(x)= 0.60554476
x=2.46875000 f(x)= 0.29337335

```

```

x=2.48437500 f(x)= 0.13716115
x=2.49218750 f(x)= 0.05904010
x=2.49609375 f(x)= 0.01997793
x=2.49804688 f(x)= 0.00044670
x=2.49902344 f(x)= -0.00931893
x=2.49853516 f(x)= -0.00443611
x=2.49829102 f(x)= -0.00199471
x=2.49816895 f(x)= -0.00077401

```

Întrebări și exerciții

1. OÎn ce cazuri se folosesc metode aproximative de determinare a soluțiilor ecuațiilor algebrice?
2. Descrieți metoda biseecției. Care sunt prioritățile ei? Dar neajunsurile?
3. Formula pentru estimarea erorii, dedusă în paragraful curent este $|\xi - a_i| < \varepsilon = |b_i - a_i|$. Exprimați diferența dintre valoarea exactă și cea calculată prin o formulă care depinde doar de extremitățile segmentului inițial și numărul de divizări realizate.
4. Descrieți algoritmul metodei biseecției.
5. Determinați prin metoda biseecției soluțiile ecuațiilor:

$$e^x - x^2 = 0 \quad \text{pe segmentul } [-1, -0.5]$$

$$x^3 - x - 1 = 0 \quad \text{pe segmentul } [1, 2]$$

$$x^3 + 3x^2 - 3 = 0 \quad \text{pe segmentul } [-3, -2]$$

$$x^5 - x - 0.2 = 0 \quad \text{pe segmentul } [1, 2]$$

- a) pentru 10, 20, 40 divizări ale segmentului inițial
 - b) cu precizia $\varepsilon = 0,001; 0,0001; 0,00001$.
 - c) în condițiile punctului precedent determinați numărul de divizări necesare pentru a obține precizia cerută.
6. Determinați prin metoda biseecției soluțiile ecuațiilor de pe intervalele separate în exercițiul 3, pagina 30.

4. METODA COARDELOR


Metoda biseției, cu toată simplitatea ei nu este efectivă în cazurile când rezultatul trebuie obținut prin un număr mic de iterații, cu o exactitate înaltă, segmentul inițial care conține soluția fiind destul de mare. În acest caz este mai potrivită divizarea segmentului în părți proporționale, proporția fiind dată de punctul de intersecție al *coardei* care unește extremitățile segmentului cu axa OX .

Fie dată funcția $f(x)$, care posedă următoarele proprietăți:


- (1) $f(x)$ continuă pe segmentul $[a, b]$ și $f(a) \times f(b) < 0$.
- (2) Pe segmentul $[a, b]$ există $f'(x) \neq 0$; $f''(x) \neq 0$ continui, și semnul lor pe $[a, b]$ este constant.

Proprietățile enumerate garantează existența soluției unice a ecuației $f(x) = 0$ pentru $x \in [a, b]$.

Metoda coardelor presupune alegerea în calitate de aproximare a soluției punctului determinat de intersecția dreptei ce trece prin punctele $(a, f(a))$ și $(b, f(b))$ cu axa OX .



Desenul 5. Aproximarea succesivă de soluția ecuației prin metoda coardelor. Extremitatea fixă - b



Desenul 6 Aproximarea succesivă de soluția ecuației prin metoda coardelor. Extremitatea fixă - a

Realizarea metodei este următoarea: se stabilește extremitatea e a segmentului $[a, b]$ prin care se va duce o serie de coarde. Această extremitate este stabilită de condiția:

$$f(e) \times f''(e) > 0.$$

Cealaltă extremitate a segmentului $[a, b]$ se consideră aproximare inițială a soluției: x_0 .

Prin punctele $(e, f(e))$ și $(x_0, f(x_0))$ se duce o coardă. Se determină punctul x_1 în care ea intersectează axa OX . Punctul x_1 este considerat următoarea aproximare a soluției.

Procesul se repetă, coarda următoare fiind dusă prin punctele $(e, f(e))$ și $(x_1, f(x_1))$. Astfel se obține șirul de aproximări $x_0, x_1, x_2, \dots, x_i, x_{i+1}, \dots, x_n \dots$ limita căruia este soluția exactă a ecuației $f(x) = 0$.

Punctele e și x_0 sînt cunoscute. Folosind ecuația dreptei ce trece prin două puncte putem determina aproximarea x_1 ($f(x_1) = 0$):

$$\frac{x - x_0}{e - x_0} = \frac{y - f(x_0)}{f(e) - f(x_0)} \Rightarrow x - x_0 = \frac{y - f(x_0)}{f(e) - f(x_0)}(e - x_0) \Rightarrow$$

$$x_1 = x_0 - \frac{f(x_0)}{f(e) - f(x_0)}(e - x_0).$$

În general, avînd calculată aproximarea x_{i-1} , putem determina următoarea aproximare

$$x_i = x_{i-1} - \frac{f(x_{i-1})}{f(e) - f(x_{i-1})}(e - x_{i-1}), \quad i = 1, 2, \dots \quad (3)$$

x_i prin formula recurentă:

Se demonstrează că șirul de valori $x_1, x_2, \dots, x_i, x_{i+1}, \dots, x_n, \dots$ calculate după formula (3) converge către soluția ξ a ecuației $f(x) = 0$.

Demonstrație* Fie $f(a) > 0; f''(a) > 0$ (Desenul 6) (Celelalte cazuri posibile se demonstrează la fel)

Deoarece $f''(a) > 0$, pe segmentul $[a, b]$ graficul funcției este inferior coardelor cu extremitățile aparținînd segmentului cercetat.

$$x_i = x_{i-1} - \frac{f(x_{i-1})}{f(a) - f(x_{i-1})}(a - x_{i-1}), \quad i = 1, 2, \dots \quad (3.a)$$

Extremitatea fixă este a . Formula (3) se transformă:

Din (3.a) rezultă că $x_i < x_{i-1} \forall i$:

$$\begin{aligned} f(x_{i-1}) &< 0, \quad i = 1, 2, \dots \\ (a - x_{i-1}) &< 0, \quad i = 1, 2, \dots \\ f(a) - f(x_{i-1}) &> 0, \quad i = 1, 2, \dots \end{aligned}$$

În urma utilizării repetate a formulei (3.a) se obține șirul $x_0 > x_1 > \dots > x_i > \dots > \xi_1 > a$.

Șirul este descrescător și mărginit inferior, deci există limita lui ξ_1 .

Trecînd la limită în formula (3.a), se obține

$$\lim_{i \rightarrow \infty} x_i = \lim_{i \rightarrow \infty} x_{i-1} - \frac{f(\lim_{i \rightarrow \infty} x_{i-1})}{f(a) - f(\lim_{i \rightarrow \infty} x_{i-1})}(a - \lim_{i \rightarrow \infty} x_{i-1})$$

sau, după înlocuire

$$\xi_1 = \xi_1 - \frac{f(\xi_1)}{f(a) - f(\xi_1)}(a - \xi_1) \Rightarrow 0 = \frac{f(\xi_1)}{f(a) - f(\xi_1)}(a - \xi_1)$$

de unde rezultă $0 = f(\xi_1)$. (numitorul fracției e diferit de 0, al doilea factor – la fel). Deci ξ_1 este soluția ecuației inițiale. $\xi_1 = \xi$, **ceea ce și trebuia de demonstrat**.

Estimarea erorii

Faptul că șirul aproximărilor succesive prin metoda coardelor converge către soluția exactă implică următoarea concluzie: cu cât mai multe iterații ale metodei vor fi realizate, cu

* Demonstrațiile propuse sunt opționale aici și în paragrafele care urmează.

atât mai bine va fi aproximată soluția. Totuși, am putea determina o formulă, care permite estimarea erorii de calcul, și, prin urmare, exactitatea³ soluției obținute.

Demonstrarea formulei de estimare a erorii. Deoarece ξ este soluția exactă a ecuației, $f(\xi) = 0$. Se adăugă $f(\xi)$ la numărătorul fracției din (3.a), apoi se estimează diferența între soluția exactă și cea calculată iterativ:

$$x_i = x_{i-1} - \frac{f(x_{i-1})}{f(a) - f(x_{i-1})}(a - x_{i-1}); \quad x_i = x_{i-1} + \frac{f(\xi) - f(x_{i-1})}{f(a) - f(x_{i-1})}(a - x_{i-1})$$

$$f(\xi) - f(x_{i-1}) = \frac{f(a) - f(x_{i-1})}{a - x_{i-1}}(x_i - x_{i-1}) \quad (4)$$

Pentru estimarea finală se va folosi **Teorema Lagrange**⁴.

Formula din teoremă se va aplica aparte pentru ambele părți a egalității (4).

$$f(\xi) - f(x_{i-1}) = (\xi - x_{i-1}) f'(r) \quad r \in [\xi, x_{i-1}];$$

$$f(a) - f(x_{i-1}) = (a - x_{i-1}) f'(r^*) \quad r^* \in [a, x_{i-1}]$$

După înlocuirea în (4), se obține:

$$(\xi - x_{i-1}) f'(r) = (x_i - x_{i-1}) f'(r^*) \Rightarrow |\xi - x_i| = \left| \frac{f'(r^*) - f'(r)}{f'(r)} \right| \times |x_i - x_{i-1}| \Rightarrow$$

$$|\xi - x_i| \leq \left| \frac{M_1 - m_1}{m_1} \right| \times |x_i - x_{i-1}|$$

Prin urmare, dacă se cere calculul soluției cu o exactitate dată ε , calculele se vor repeta conform formulei (3.a) până la obținerea inegalității

$$\left| \frac{M_1 - m_1}{m_1} \right| \times |x_i - x_{i-1}| \leq \varepsilon \quad (5)$$

unde M_1 și m_1 sînt corespunzător marginea superioară și inferioară a $f'(x)$ pe $[a, b]$.

Algoritmul de calcul

Aplicarea metodei coardelor necesită o cercetare prealabilă a funcției $f(x)$, pentru stabilirea extremității fixe, din care vor fi duse coardele. Numărul n de aproximări succesive ale soluției poate fi indicat în enunțul problemei, sau determinat de o condiție. În ambele cazuri calculul se realizează conform formulei (3.a). Condiția de oprire în primul caz va fi aplicarea repetată de n ori a formulei (3.a); în cel de al doilea – îndeplinirea condiției (5).

Determinarea extremității fixe. Dacă descrierea analitică a $f'(x)$ nu este indicată în enunțul problemei, urmează să fie dedusă, folosind pentru aceasta expresia pentru $f(x)$. Totuși, deducerea formulei pentru $f'(x)$ poate fi omisă prin următoarea procedură:

Se determină semnul $f(x)$ în punctul c de intersecție al dreptei ce trece prin punctele $(a, f(a))$ și $(b, f(b))$ cu axa OX . **Fixă va fi acea extremitate e , pentru care se îndeplinește condiția:**

³ precizia

⁴ Fie $f: [a, b] \rightarrow \mathbb{R}$, continuă și derivabilă pe $[a, b]$. Atunci $\exists c \in (a, b)$ astfel încât:
 $f(b) - f(a) = (b - a) f'(c)$

$$f(e) \times f(c) < 0.$$

Algoritmul de calcul pentru un număr prestabilit n de aproximări succesive:

Pentru a realiza acest algoritm este suficient să se cunoască descrierea analitică a $f(x)$

Pas 1 Determinarea extremității fixe e și a aproximării x_0 ; $i \leftarrow 0$.

Pas 2 Calculul x_{i+1} conform formulei $x_{i+1} = x_i - \frac{f(x_i)}{f(e) - f(x_i)} (e - x_i)$;

Pas 3 Dacă $i+1 = n$ atunci SFÂRȘIT,
în caz contrar $i \leftarrow i+1$, și se revine la **pas 2**

Algoritmul de calcul pentru o exactitate ε dată:

Deoarece în formula de estimare a erorii figurează mărimile M_1 și m_1 , în cazul când valorile lor nu sunt indicate în enunțul problemei, este necesară o preprocesare matematică pentru stabilirea M_1 și m_1 . Suplimentar sunt necesare descrierile analitice pentru $f(x)$, $f'(x)$.

Pas 1 Determinarea extremității fixe e și a aproximării x_0 ; $i \leftarrow 0$.

Pas 2 Calculul x_{i+1} conform formulei $x_{i+1} = x_i - \frac{f(x_i)}{f(e) - f(x_i)} (e - x_i)$;

Pas 3 Dacă $\left| \frac{M_1 - m_1}{m_1} \right| \times |x_i - x_{i-1}| \leq \varepsilon$ atunci SFÂRȘIT,
în caz contrar $i \leftarrow i+1$, și se revine la **pas 2**

Exemplul 1: Fie dată funcția $f(x) = \ln(x \sin x)$. Se cere să se calculeze soluția aproximativă a ecuației $f(x) = 0$ pe segmentul $[0,5; 1,5]$ pentru 15 aproximări succesive, utilizând metoda coardelor.

Preprocesarea matematică – nu este necesară.

Programul. Deoarece numărul de aproximări succesive este fixat, iar extremitățile segmentului cunoscute, atribuirile respective vor fi realizate direct în corpul programului.

```

program cn005;
var
  a,b,e,c,x: real;
  n,i: integer;

function f(x:real):real;
begin
  f:=ln(x*sin(x));
end;

begin
  a:=0.5; b:=1.5; n:=15;
  {determinarea extremitatii fixe e si a aproximarii initiale x0}
  c:=a-(f(a))/(f(b)-f(a))*(b-a);
  if f(c)*f(a)>0 then begin e:=b; x:=a; end
  else begin e:=a; x:=b; end;

  {calculul iterativ al solutiei}
  for i:=1 to n do
    begin

```

```

        x:= x-(f(x))/(f(e)-f(x))*(e-x);
        writeln(t,x:10:8, ' ', f(x):12:8);
    end;
end.

```

Rezultate:

```

i= 1 x=1.27995775 f(x)= 0.20392348
i= 2 x=1.18251377 f(x)= 0.09028687
i= 3 x=1.14193561 f(x)= 0.03779857
i= 4 x=1.12538555 f(x)= 0.01546570
i= 5 x=1.11868644 f(x)= 0.00626938
i= 6 x=1.11598267 f(x)= 0.00253191
i= 7 x=1.11489268 f(x)= 0.00102097
i= 8 x=1.11445346 f(x)= 0.00041145
i= 9 x=1.11427651 f(x)= 0.00016577
i= 10 x=1.11420523 f(x)= 0.00006678
i= 11 x=1.11417651 f(x)= 0.00002690
i= 12 x=1.11416494 f(x)= 0.00001084
i= 13 x=1.11416028 f(x)= 0.00000437
i= 14 x=1.11415841 f(x)= 0.00000176
i= 15 x=1.11415765 f(x)= 0.00000071

```

Exemplul 2: Fie dată funcția $f(x) = x^4 - 3x^2 + 7,5x - 1$. Se cere să se calculeze soluția aproximativă a ecuației $f(x) = 0$ pe segmentul $[-0,5; 0,5]$ cu exactitatea $\varepsilon = 0.0001$, utilizând metoda coardelor. Pentru funcția dată pe segmentul $[-0,5; 0,5]$ M_1 și m_1 sînt respectiv egale cu 10 și 5.

Preprocesarea matematică nu este necesară, deoarece mărimile necesare pentru calculul soluției cu exactitatea cerută sînt indicate în enunț.

Program

Pentru simplitate atribuirile necesare vor fi realizate direct în corpul programului.

```

program cn006;
var
    Msup,minf,a,b,e,x,xnou,xvechi,eps: real;

function f(x:real):real;
begin
    f:=sqr(sqr(x))-3*sqr(x)+7.5*x-1;
end;

begin
    a:=-0.5; b:=0.5; eps:=0.0001;
    Msup:=10; minf:=5;

    {determinarea extremitatii fixe si a aproximarii initiale}
    x:=a-(f(a))/(f(b)-f(a))*(b-a);
    if f(x)*f(a)>0 then begin e:=b; xnou:=a; end
    else begin e:=a; xnou:=b; end;

    {calculul iterativ al solutiei}

```

```

repeat
    xvechi:=xnou;
    xnou:= xvechi-(f(xvechi))/(f(e)-f(xvechi))*(e-xvechi);
    writeln(' x=',xnou:10:8, ' f(x)=',f(xnou):12:8);
until abs((Msup-minf)/minf*(xnou-xvechi))<eps;
end.

```

Rezultate:

```

x=0.22500000 f(x)= 0.53818789
x=0.15970438 f(x)= 0.12191694
x=0.14523719 f(x)= 0.02644238
x=0.14211461 f(x)= 0.00567781
x=0.14144482 f(x)= 0.00121650
x=0.14130134 f(x)= 0.00026052
x=0.14127062 f(x)= 0.00005579

```

Exemplul 3: Fie dată funcția $f(x) = x[\sin(\ln(x)) - \cos(\ln(x))]$. Să se calculeze soluția aproximativă a ecuației $f(x) = 0$ pe segmentul $[2; 3]$ cu exactitatea $\varepsilon = 0.0001$, utilizând metoda coardelor.

Preprocesarea matematică

Se determina $f'(x)$.

$$f(x) = x[\sin(\ln(x)) - \cos(\ln(x))];$$

$$f'(x) = 2\sin(\ln(x)).$$

Pe segmentul $[2, 3]$ funcția $\ln(x)$ este pozitivă, crescătoare, $\ln(3) < \frac{\pi}{2}$. Prin urmare $2\sin(\ln(x))$ va fi de asemenea pozitivă, crescătoare pe $[2, 3]$ și m_1 , M_1 vor fi respectiv $2\sin(\ln(2))$ și $2\sin(\ln(3))$.

Program. Atribuirile necesare se vor efectua direct în corpul programului.

```

program cn007;
var
    Msup,minf,a,b,e,x,xnou,xvechi,eps: real;

function f(x:real):real;
begin
    f:=x*(sin(log(x))-cos(log(x)));
end;

begin
    a:=2; b:=3; eps:=0.00001;
    Msup:=2*sin(log(3)); minf:=2*sin(log(2));
    x:=a-(f(a))/(f(b)-f(a))*(b-a);
    if f(x)*f(a)>0 then begin e:=b; xnou:=a; end
    else begin e:=a; xnou:=b; end;
    repeat
        xvechi:=xnou;
        xnou:= xvechi-(f(xvechi))/(f(e)-f(xvechi))*(e-xvechi);
        writeln('x=',xnou:10:8, ' f(x)=',f(xnou):12:8);
    until abs((Msup-minf)/minf*(xnou-xvechi))<eps;
end.

```

Rezultate:

x=2.16619279 f(x)= -0.03806866
x=2.18978743 f(x)= -0.00493538
x=2.19283483 f(x)= -0.00062958
x=2.19322338 f(x)= -0.00008014
x=2.19327284 f(x)= -0.00001020
x=2.19327913 f(x)= -0.00000130

Întrebări și exerciții

1. Explicați esența metodei coardelor. Descrieți metoda grafic.
2. Cum depinde extremitatea fixă de semnele derivatei 1 și 2?
3. Demonstrați, că la alegerea corectă a extremității fixe, șirul aproximărilor obținute prin metoda coardelor converge către soluția exactă a ecuației.
4. Care este exactitatea metodei, cum poate fi obținută?
5. Descrieți procesul de determinare a extremității fixe. Cum poate fi omis calculul $f''(x)$?
6. Descrieți algoritmul metodei coardelor pentru un număr fix de iterații și pentru o exactitate dată.
7. Determinați soluțiile ecuațiilor, utilizând metoda coardelor:
 - a. $x^3 - 0.2x^2 + 0.2x + 1.2 = 0$ pe $[1, 2]$
 - b. $5x^3 - 20x + 3 = 0$ pe $[0, 1]$
 - c. $e^x - x^2 = 0$ pe $[-1, 0]$
8. Separați și determinați mai apoi soluțiile ecuațiilor, utilizând metoda coardelor:
 - a. $\operatorname{tg}(0.55x+0.1)=x^2$;
 - b. $x^3-0.2x^2+0.5x+1.5=0$.

5. METODA NEWTON

Fie dată funcția $f(x)$, care posedă următoarele proprietăți:

- (1) $f(x)$ continuă pe segmentul $[a, b]$ și $f(a)f(b) < 0$.
- (2) Pe segmentul $[a, b]$ există $f'(x) \neq 0$, $f''(x) \neq 0$, continui, și semnul lor pe $[a, b]$ este constant.

La fel ca în paragrafele precedente, scopul este de a rezolva ecuația $f(x) = 0$ pentru $x \in [a, b]$. Se va încerca rezolvarea problemei prin trasarea consecutivă a unor tangente la graficul funcției, prima dintre ele fiind construită prin extremitatea E_0 a segmentului $[a, b]$, extremitate, pentru care se respectă condiția:

$$f(x_0) \times f''(x_0) > 0$$

(x_0 - abscisa extremității).

Fie că tangenta cu numărul i intersectează axa OX în punctul x_i . Următoarea tangentă ($i+1$) va fi trasată prin punctul E_{i+1} cu coordonatele $(x_i, f(x_i))$ și va intersecta axa absciselor în punctul x_{i+1} . Șirul de valori $x_0, x_1, x_2, \dots, x_i, x_{i+1}, \dots$ va converge către soluția ecuației $f(x) = 0$. Această metodă de calcul a soluției ecuației $f(x)=0$ este numită **metoda tangen-telor** sau **Newton**, după numele matematicianului, care a introdus-o.

Geometric, metoda este ilustrată în desenul 7a.

Pentru a calcula valorile $x_1, x_2, \dots, x_i, \dots$ se va folosi ecuația tangentei la funcție, ce trece printr-un punct dat:


$$y - f(x_i) = f'(x_i)(x - x_i) \quad (3)$$

În caz general ecuația (3) reprezintă tangenta la funcția $f(x)$, ce trece prin punctul $(x_i, f(x_i))$. Ea va intersecta axa absciselor în punctul cu coordonatele $(x_{i+1}, 0)$. Ținând cont de faptul, că în punctul de intersecție a tangentei cu axa OX valoarea ordonatei este 0, se obține:


$$x_{i+1} = x_i - \frac{f(x_i)}{f'(x_i)} \quad (4)$$

Utilizarea în calitate de punct inițial a extremității pentru care nu se respectă condiția $f(x_0)f''(x_0) > 0$ poate duce la construirea tangențelor care intersectează axa OX în afara segmentului $[a, b]$ (desenul 7b). În consecință, se trece pe un alt interval, pe care se determină o altă soluție (dacă ea există), sau se generează erori de execuție.

Regula de selectare în calitate de punct inițial a extremității x_0 pentru care se îndeplinește condiția $f(x_0)f''(x_0) > 0$ este una generală și e bazată pe următoarea teoremă:



Desenul 7a: convergența șirului de valori $x_0, x_1, x_2, \dots, x_i, x_{i+1}, \dots$ către soluția exactă ξ .



Desenul 7b: tangenta dusă în extremitatea pentru care nu se respectă condiția $f(x_0)f''(x_0) > 0$ poate părăsi segmentul pe care se realizează căutarea soluției.

Fie funcția $f(x)$ cu proprietățile (1), (2). Dacă în calitate de aproximare inițială a soluției se ia extremitatea $x_0 : f(x_0)f'(x_0) > 0$, atunci soluția unică a ecuației $f(x) = 0$ pe segmentul $[a, b]$, poate fi calculată cu orice grad de exactitate, utilizând formula (4) (*fără demonstrație*)

Estimarea erorii

Pentru estimarea erorii la aproximarea cu numărul $i+1$ (în punctul x_{i+1}), se va folosi pentru funcția $f(x)$ formula Taylor cu termen complementar în forma lui Lagrange:

$$f(x) = f(a) + f'(a)(x-a) + \dots + \frac{f^{(n)}(a)}{n!}(x-a)^n + \frac{f^{(n+1)}(a)}{(n+1)!}(x-a)^{n+1}$$

unde a este un număr real, iar c este un punct situat între a și x .

Astfel, pentru $x=x_{i+1}$ și $a=x_i$, $n=1$, formula Taylor capătă forma:

$$f(x_{i+1}) = f(x_i) + f'(x_i)(x_{i+1} - x_i) + \frac{f''(\xi_i)}{2}(x_{i+1} - x_i)^2; \quad \xi_i \in (x_i, x_{i+1}). \quad (5)$$

Conform (4), $f(x_i) + f'(x_i)(x_{i+1} - x_i) = 0$,

$$\text{Prin urmare } |f(x_{i+1})| \leq \frac{1}{2} M_2 (x_{i+1} - x_i)^2; \quad M_2 = \sup_{x \in [a, b]} |f''(x)|.$$

Prin înlocuire în formula generală a erorii (§ 3.2), se obține:

$$|\xi - x_{i+1}| \leq \frac{M_2}{2m_1} (x_{i+1} - x_i)^2. \quad (6)$$

Algoritmul

Numărul de aproximări succesive în procesul de calcul poate fi stabilit apriori sau determinat de o condiție. În orice caz este necesar să se stabilească mai întâi extremitatea segmentului, care va servi drept aproximare inițială. Calculul aproximării următoare se realizează în ambele cazuri conform formulei (4). Condiția de oprire a calculelor va fi în primul caz generarea aproximării cu indicele cerut; în cel de al doilea – îndeplinirea condiției (6).

Algoritmul de calcul pentru un număr dat de aproximări succesive:

Pentru a realiza acest algoritm este suficient să fie cunoscute descrierile analitice pentru $f(x)$, $f'(x)$ și $f''(x)$. Dacă descrierea $f'(x)$ nu este indicată în enunț, urmează să fie calculată. Calculul $f''(x)$ poate fi omis prin folosirea unei proceduri similare celei de determinare a extremității fixe pentru metoda coardelor: extremitatea fixă pentru metoda coardelor va fi în același timp aproximarea inițială pentru metoda tangentelor.

Pas 1 Determinarea extremității x_0 din care încep aproximările succesive ale soluției. $i \leftarrow 0$.

Pas 2 Se calculează x_{i+1} conform formulei $x_{i+1} = x_i - \frac{f(x_i)}{f'(x_i)}$

Pas 3 Dacă $i+1 = n$ atunci SFÂRȘIT, în caz contrar $i \leftarrow i+1$, apoi **pas 2**

Algoritmul de calcul pentru o exactitate ε dată:

În formula de estimare a erorii figurează mărimile M_2 și m_1 . În cazul când valorile lor nu sunt indicate în enunțul problemei, este necesară o preprocesare matematică pentru stabilirea M_2 și m_1 . Suplimentar sunt necesare descrierile analitice pentru $f(x)$, și $f'(x)$.

Pas 1 Determinarea extremității x_0 din care încep aproximările succesive ale soluției. $i \leftarrow 0$.

Pas 2 Se calculează x_{i+1} conform formulei $x_{i+1} = x_i - \frac{f(x_i)}{f'(x_i)}$

Pas 3 Dacă $\frac{M_2}{2m_1}(x_{i+1} - x_i)^2 \leq \varepsilon$ atunci SFÂRȘIT,

în caz contrar $i \leftarrow i+1$, și se revine la **pas 2**

Exemplul 15

Fie dată funcția $f(x) = x^3 - 2x^2 + x - 3$. Se cere să se calculeze soluția aproximativă a ecuației $f(x) = 0$ pe segmentul $[2; 15]$ pentru 10 aproximări succesive, utilizând metoda Newton.

Preprocesarea matematică. Se determină $f'(x)$.

$$f(x) = x^3 - 2x^2 + x - 3; \quad f'(x) = 3x^2 - 4x + 1.$$

Programul. Deoarece numărul de aproximări succesive este fixat, iar extremitățile segmentului cunoscute, atribuiriile necesare se vor realiza direct în corpul programului.

```
program cn008;
var a, b, x, c : real;
    i, n: integer;

function f(z:real):real;
begin f:=z*z*z-2*z*z+z-3; end;

function fd1(z:real):real;
begin fd1:=3*z*z-4*z+1; end;

begin
  a:=2.1; b:=15; n:=10; i:=0;
  c:=a-(f(a))/(f(b)-f(a))*(b-a);
  if f(c)*f(a)>0 then x:=a else x:=b;
  while i<n do
    begin
      i:=i+1;
      x:=x-f(x)/fd1(x);
      writeln('i=', i:2, ' x=', x:15:12, ' f=', f(x):15:12);
    end;
  end.
```

Rezultate:

```
i= 1 x= 10.23214285700 f=869.11072454000
i= 2 x= 7.06207637180 f=256.52261987000
i= 3 x= 4.96579746180 f= 75.09982542600
i= 4 x= 3.60317646350 f= 21.41702511300
i= 5 x= 2.76447507070 f= 5.60684004000
i= 6 x= 2.32879157830 f= 1.11191715150
i= 7 x= 2.18900944530 f= 0.09469778945
i= 8 x= 2.17470302090 f= 0.00093182281
i= 9 x= 2.17455942470 f= 0.00000009329
```

⁵ Pentru toate exemplele și exercițiile propuse se presupune îndeplinirea condițiilor (1) și (2) pentru $f(x)$.

i=10 x= 2.17455941030 f= 0.00000000001

Exemplul 2. Fie dată funcția $f(x) = \cos^2(x) - \frac{x}{4}$. Se cere să se calculeze soluția aproximativă a ecuației $f(x) = 0$ pe segmentul $[2,4; 3]$ cu exactitatea $\varepsilon = 0.0001$, utilizând metoda Newton. Pentru funcția dată pe segmentul $[2,4; 3]$ M_2 și m_1 sunt respectiv egale cu 2 și 0.03.

Preprocesarea matematică.

$$f(x) = \cos^2(x) - \frac{x}{4}; \quad f'(x) = -\sin(2x) - \frac{1}{4}; \quad f''(x) = -2\cos(2x).$$

Program. Deoarece ε este dat, extremitățile segmentului și valorile M_2 , m_1 - cunoscute, atribuirile vor fi realizate direct în program.

```

program cn009;
var a, b, xn, xv, M2, m1, e, c : real;
    i, n: integer;

function f(z:real):real;
begin
    f:=cos(z)*cos(z)-z/4;
end;

function fd1(z:real):real;
begin
    fd1:=-sin(2*z)-1/4;
end;

begin
    a:=2.4; b:=3; M2:=2; m1:=0.03; e:=0.0001;
    c:=a-(f(a))/(f(b)-f(a))*(b-a);
    if f(c)*f(a)>0 then begin
        xn:=a; xv:=b;
    end
    else begin
        xn:=b; xv:=a;
    end;
    while M2*sqr(xn-xv)/(2*m1)>e do
    begin
        xv:=xn;
        xn:=xv-f(xv)/fd1(xv);
        writeln(' x=',xn:15:12, ' f=',f(xn):15:12);
    end;
end.

```

Rezultate:

x= 2.47538619170 f= -0.00078052066
x= 2.47646766320 f= -0.00000027700
x= 2.47646804730 f= 0.00000000000

Exemplul 3: Fie dată funcția $f(x) = \sin^2(x) - \frac{x}{2}$. Se cere să se calculeze soluția aproximativă a ecuației $f(x) = 0$ pe segmentul $[0,5; 0,7]$ cu exactitatea $\varepsilon = 0.00001$, utilizând metoda Newton.

Preprocesarea matematică

$$f(x) = \sin^2(x) - \frac{x}{2}; \quad f'(x) = \sin(2x) - \frac{1}{2}; \quad f''(x) = 2\cos(2x).$$

Pe segmentul $[0,5; 0,7]$ funcția $\sin(2x)-0,5$ este pozitivă, crescătoare. Prin urmare m_1 va fi atins în $x = 0,5$ și va fi egal cu $\sin(1) - 0,5$. M_2 nu va depăși valoarea 2. Această afirmație rezultă din proprietățile funcției $\cos(x)$.

Program

```

program cn010;
var a, b, xn, xv, M2, m1, e, c : real;
    i, n: integer;

function f(z:real):real;
begin f:=sin(z)*sin(z)-z/2; end;

function fd1(z:real):real;
begin fd1:=sin(2*z)-1/2; end;

begin
  a:=0.5; b:=0.7; M2:=2; m1:=sin(1)-0.5; e:=0.000001;
  c:=a-(f(a))/(f(b)-f(a))*(b-a);
  if f(c)*f(a)>0 then begin
    xn:=a; xv:=b;
    end
  else begin
    xn:=b; xv:=a;
    end;
  while M2*sqr(xn-xv)/(2*m1)>e do
  begin
    xv:=xn;
    xn:=xv-f(xv)/fd1(xv);
    writeln(' x=',xn:15:12, ' f=',f(xn):15:12);
  end;
end.

```

Rezultate:

```

x= 0.56606969881 f= 0.00460318274
x= 0.55471287050 f= 0.00005566154
x= 0.55457211314 f= 0.00000000882

```

Întrebări și exerciții

1. Descrieți sensul geometric al metodei Newton.
2. Cum poate fi stabilită aproximarea inițială a metodei?
3. Demonstrați, că la alegerea corectă a punctului inițial, șirul aproximărilor obținute prin metoda Newton converge către soluția exactă a ecuației.
4. Care este exactitatea metodei, cum poate fi obținută?
5. Determinați soluțiile ecuațiilor pentru 5 iterații, utilizând metoda Newton:

$$4x^4 + 8x^3 - 3x^2 - 7x + 3 = 0 \text{ pe } [-1,7; -1,58], [-1,53; -1,4], [0,4; 0,52], [0,58; 0,8]$$

$$\sqrt{x + \sqrt{x + \sqrt{x}}} - 12 = 0 \quad [100; 150]$$

$$(2 - x^2) \cos(x) + 2x \sin(x) = 0 \quad [-4, 5; -4] \quad [4; 4, 5]$$

$$\ln\left(\frac{1}{x} + \ln\left(\frac{1}{x} + \ln\frac{1}{x}\right)\right) = 0 \quad [0, 1; 0, 5]$$

6. Calculați soluțiile ecuațiilor, utilizând metoda Newton, pentru $\varepsilon = 0.00001$:

$$\text{a) } 2\cos 2(x) - e^{x/2} = 0 \quad [0, 1; 0, 74] \quad M_2=4, \quad m_1=0.5$$

$$\text{b) } x^5 - 4x + 9 = 0 \quad [-2; -1] \quad M_2=150, \quad m_1 = 1$$

7. Separați și calculați mai apoi soluțiile ecuațiilor, utilizând metoda Newton, pentru $\varepsilon=0.00001$:

$$x^5 - 80x^2 + 89 = 0$$

$$e^x - x^2 = 0$$

6. METODA MIXTĂ⁶

În cazul rezolvării unei ecuații de forma $f(x) = 0$ atât metoda coardelor, cât și metoda tangentelor sînt mai efective decît metoda bisecției. Totuși, un neajuns al lor este necesitatea calculării prealabile ale unor valori auxiliare, cum ar fi supremul și infimul derivatei de ordinul 1 și 2 a funcției. În același timp, metoda bisecției, deși mai lentă, permite localizarea soluției exacte în interiorul unui segment $[a_i, b_i]$, extremitățile căruia sunt cunoscute. Pentru această metodă eroarea de calcul a soluției $\leq |a_i - b_i|$. Apare ideea de a îmbina eficacitatea metodei coardelor și a tangentelor cu simplitatea evaluării erorii preluată de la metoda bisecției.

Fie dată funcția $f(x)$, care posedă următoarele proprietăți:

1. $f(x)$ continuă pe segmentul $[a, b]$ și $f(a)f(b) < 0$.
2. Pe segmentul $[a, b]$ există $f'(x) \neq 0$, $f''(x) \neq 0$, continue, și semnul lor pe $[a, b]$ este constant.

Pentru a rezolva ecuația $f(x) = 0$ poate fi aplicată atât metoda coardelor, cât și metoda tangentelor. Se observă ușor următoarea legătură: *dacă extremitatea a a segmentului $[a, b]$ este extremitatea fixă pentru metoda coardelor, tot ea este punctul de start în cazul aplicării metodei tangentelor.*


În acest caz șirul de aproximări succesive $x_0=b, x_1, x_2, \dots, x_n$ obținute prin metoda coardelor va converge către soluția exactă pornind de la extremitatea b . Șirul de aproximări succesive $z_0=a, z_1, z_2, \dots, z_n$ obținute prin metoda tangentelor va converge și el către soluția exactă, dar pornind de la extremitatea a . Soluția exactă se va afla întotdeauna în interiorul segmentului $[z_i, x_i]$. Prin urmare eroarea de calcul a soluției după i iterații nu va depăși lungimea segmentului $[z_i, x_i]$.

Aceeași legătură se observă și în cazul când extremitatea b este fixă pentru metoda coardelor. În acest caz șirul aproximărilor succesive $x_0=a, x_1, x_2, \dots, x_n$ obținute prin metoda coardelor va converge către soluția exactă pornind de la extremitatea a , iar șirul de aproximări succesive $z_0=b, z_1, z_2, \dots, z_n$ obținute prin metoda tangentelor va converge către soluția exactă pornind de la extremitatea b . (desenul 8)

Din observațiile anterioare rezultă metoda mixtă: se determină prima ($i = 1$) aproximare z_1 prin metoda tangentelor și prima aproximare x_1 prin metoda coardelor. Atât timp cât $\Delta (\Delta = |x_i - z_i|)$ este mai mare decît precizia de calcul cerută în condițiile problemei se calculează următoarea pereche de aproximări. În calitate de soluție calculată poate fi luat mijlocul ultimului segment determinat de aproximările x_i, z_i . Pentru calculul aproximărilor se vor folosi formulele deja cunoscute:

$$z_{i+1} = z_i - \frac{f(z_i)}{f'(z_i)} \quad (\text{metoda tangentelor})$$

$$x_{i+1} = x_i - \frac{f(x_i)}{f(e) - f(x_i)} \quad (e = x_i); \quad e - \text{extremitatea fixă (metoda coardelor)}$$



Desenul 8. Procesul de apropiere de soluția exactă are loc concomitent din ambele părți

⁶ Opțional

O optimizare considerabilă a metodei se obține în cazul când în calitate de punct fix pentru metoda coardelor se ia ultima aproximare obținută prin metoda tangentelor (desenul 9)

În acest caz formulele de calcul capătă forma:

$$z_{i+1} = z_i - \frac{f(z_i)}{f'(z_i)} \quad (\text{metoda tangentelor})$$

$$x_{i+1} = x_i - \frac{f(x_i)}{f(z_{i+1}) - f(x_i)} (z_{i+1} - x_i);$$

z_{i+1} – extremitatea fixă (metoda coardelor)

Estimarea erorii

Eroarea de calcul Δ_i la iterația cu numărul i este calculată după formula: $\Delta_i = |x_i - z_i|$

Algoritmul de calcul

Aplicarea metodei mixte necesită o cercetare prealabilă a funcției $f(x)$, pentru stabilirea expresiei $f'(x)$, necesare pentru partea care ține de metoda Newton. La fel este necesară stabilirea extremității fixe e .


Algoritmul pentru precizia de calcul ε

1. Se determină punctul de start z_0 pentru metoda Newton: $c \leftarrow a - (f(a))/(f(b) - f(a))(b - a)$;
Dacă $f(c)f(a) > 0$ atunci $z_0 \leftarrow b$; $x_0 \leftarrow a$; în caz contrar $z_0 \leftarrow a$; $x_0 \leftarrow b$;
2. $i \leftarrow 0$
3. Se determină $z_{i+1} = z_i - \frac{f(z_i)}{f'(z_i)}$
4. Se determină $x_{i+1} = x_i - \frac{f(x_i)}{f(z_{i+1}) - f(x_i)} (z_{i+1} - x_i)$;
5. $i \leftarrow i + 1$
6. Dacă $|z_i - x_i| > \varepsilon$ se revine la pasul 3, în caz contrar se trece la 7.
7. În calitate de soluție calculată s se ia $s = \frac{x_i + z_i}{2}$. Sfârșit.

Exemplul 1: Fie dată funcția $f(x) = x\sqrt{x^2+1} - 7$. Se cere să se calculeze soluția aproximativă a ecuației $f(x) = 0$ pe segmentul $[2; 6]$ cu precizia $\varepsilon = 0.00001$, utilizând metoda mixtă.

Preprocesarea matematică

$$f'(x) = \left(x(x^2 + 1)^{\frac{1}{2}} \right)' = (x^2 + 1)^{\frac{1}{2}} + x \left((x^2 + 1)^{\frac{1}{2}} \right)' = (x^2 + 1)^{\frac{1}{2}} + x^2 (x^2 + 1)^{-\frac{1}{2}} = \frac{2x^2 + 1}{\sqrt{x^2 + 1}}$$



Desenul 9. Optimizarea metodei mixte

Program

```

program cn011;
var
    a,b,c,x,z,e : real;

function f(x:real):real;
begin
    f:=x*sqrt(1+x*x)-7;
end;

function fd1(x:real):real;
begin
    fd1:=(2*x*x+1)/sqrt(1+x*x);
end;

begin
    a:=2; b:=6; e:=0.0001;
    c:=a-(f(a))/(f(b)-f(a))*(b-a);
    if f(c)*f(a)>0 then
        begin
            z:=b; x:=a;
        end
    else
        begin
            z:=a; x:=b;
        end;
    while abs(z-x)>e do
        begin
            z:=z-f(z)/fd1(z);
            x:= x-(f(x))/(f(z)-f(x))*(z-x);
            writeln ('z=',z:9:5,' f(z)=',f(z):9:5, ' x=', x:9:5, '
f(x)=', f(x):9:5);
        end;
        writeln((z+x)/2:9:5);
    end.

```

Rezultate

```

z= 3.54218 f(z)= 6.03747 x= 2.45514 f(x)= -0.49146
z= 2.69058 f(z)= 0.72307 x= 2.55041 f(x)= -0.01326
z= 2.55649 f(z)= 0.01787 x= 2.55300 f(x)= -0.00001
z= 2.55301 f(z)= 0.00001 x= 2.55300 f(x)= -0.00000
2.55301

```

Întrebări și exerciții

1. Descrieți metoda mixtă pentru rezolvarea ecuațiilor algebrice și transcendente. Enumerați prioritățile ei.
2. Expuneți formulele de calcul, utilizate în metoda mixtă. Din care metode au fost preluate ele?
3. Cum poate fi optimizată metoda mixtă?
4. Care este exactitatea metodei, cum poate fi obținută?
5. Descrieți algoritmul metodei pentru un număr fix de iterații. Desenați schema logică a algoritmului.
6. Rezolvați, utilizând metoda mixtă, ecuațiile propuse în paragrafele 3.1 – 3.5